

Lesson 16

Arc Length

Initializations

```
> restart;  
with(student):
```

16.1 Arc Length

Examples

The arc length L of the curve $y = f(x)$, $a \leq x \leq b$ is given by

$$L = \int_a^b \sqrt{1 + (f'(x))^2} dx$$

Mathematical details will be provided in class.

Example 16.1.1

Compute the arc length of the curve $f(x) = \frac{x^3}{2} + \frac{1}{2x}$, $1 \leq x \leq 2$.

Solution

Use the formula for arc length given at the top of this worksheet.

```
> f:=x->x^3/6+1/(2*x);
```

$$f := x \rightarrow \frac{1}{6}x^3 + \frac{1}{2x} \quad (2.1.1.1)$$

```
> e1:=Int(sqrt(1+D(f)(x)^2), x=1..2);
```

$$e1 := \int_1^2 \sqrt{1 + \left(\frac{1}{2}x^2 - \frac{1}{2x^2}\right)^2} dx \quad (2.1.1.2)$$

```
> e2:=simplify(e1);
```

$$e2 := \frac{1}{2} \int_1^2 \frac{x^4 + 1}{x^2} dx \quad (2.1.1.3)$$

```
> arclength:=value(e2);  
evalf(arclength);
```

$$\text{arclength} := \frac{17}{12} \\ 1.416666667 \quad (2.1.1.4)$$

>

Example 16.1.2

Find the arc length of the curve $f(x) = \frac{x^2}{2} + \frac{\ln x}{4}$, $2 \leq x \leq 4$.

Solution

Use the formula for arc length given at the top of this worksheet.

> `f:=x->x^2/2+ln(x)/4;`

$$f := x \rightarrow \frac{1}{2} x^2 + \frac{1}{4} \ln(x) \quad (2.1.2.1)$$

> `e1:=Int(sqrt(1+D(f)(x)^2), x=2..4);`

$$e1 := \int_2^4 \sqrt{1 + \left(x + \frac{1}{4x}\right)^2} dx \quad (2.1.2.2)$$

> `e2:=simplify(e1);`

$$e2 := \frac{1}{4} \int_2^4 \frac{\sqrt{24x^2 + 16x^4 + 1}}{x} dx \quad (2.1.2.3)$$

It should be clear that this integral can be evaluated by applying the substitution $u = x^2$ followed by a trigonometric substitution. We simply use Maple's power to evaluate this integral.

> `e3:=value(e2);`

$$\begin{aligned} e3 := & \frac{1}{8} \sqrt{4481} + \frac{1}{16} \ln(\sqrt{4481} - 1) - \frac{1}{16} \ln(1 + \sqrt{4481}) \\ & - \frac{1}{16} \ln(67\sqrt{4481} + 13443) + \frac{3}{16} \ln(67\sqrt{4481} + 4481) \\ & - \frac{3}{16} \ln(67\sqrt{4481} - 4481) + \frac{1}{16} \ln(-67\sqrt{4481} + 13443) \\ & - \frac{1}{8} \sqrt{353} - \frac{1}{16} \ln(\sqrt{353} - 1) + \frac{1}{16} \ln(1 + \sqrt{353}) \\ & + \frac{1}{16} \ln(19\sqrt{353} + 1059) - \frac{3}{16} \ln(19\sqrt{353} + 353) \\ & + \frac{3}{16} \ln(19\sqrt{353} - 353) - \frac{1}{16} \ln(-19\sqrt{353} + 1059) \end{aligned} \quad (2.1.2.4)$$

Looking at this result, one starts to appreciate the value of a decimal approximation.

> `arclength_f:=evalf(e3);`

$$arclength_f := 6.498799194 \quad (2.1.2.5)$$

Alternatively this result can be obtained by numeric integration of

$$\int_2^4 \sqrt{1 + \left(x + \frac{1}{4x}\right)^2} dx.$$

```
> e1;
check:=evalf(e1);
```

$$\int_2^4 \sqrt{1 + \left(x + \frac{1}{4x}\right)^2} dx$$

```
check := 6.498799179
```

(2.1.2.6)

```
>
```

Example 16.1.3

Compute the arc length of the curve $f(x) = x \sin x$, $0 \leq x \leq 1$.

Solution

Integrals representing arc length, frequently cannot be evaluated exactly. This holds true even for some of the most elementary functions. This example represents one of those cases.

```
> f:=x->x*sin(x);
```

$$f := x \rightarrow x \sin(x)$$

(2.1.3.1)

```
> e1:=Int(sqrt(1+D(f)(x)^2), x=0..1);
```

$$e1 := \int_0^1 \sqrt{1 + (\sin(x) + x \cos(x))^2} dx$$

(2.1.3.2)

```
> value(e1);
```

$$\int_0^1 \sqrt{1 + (\sin(x) + x \cos(x))^2} dx$$

(2.1.3.3)

Maple leaves the integral unevaluated. Numeric integration offers the only solution. First, we use Maple's default integration routine.

```
> L:=evalf(e1);
```

$$L := 1.351336280$$

(2.1.3.4)

Then we apply something as simple as the Simpson's rule with 20 sub-intervals.

```
> e2:=simpson(sqrt(1+D(f)(x)^2), x=0..1, 20);
```

$$e2 := \frac{1}{60} + \frac{1}{60} \sqrt{1 + (\sin(1) + \cos(1))^2} + \frac{1}{15} \sum_{i=1}^{10}$$

(2.1.3.5)

$$\sqrt{1 + \left(\sin\left(\frac{1}{10}i - \frac{1}{20}\right) + \left(\frac{1}{10}i - \frac{1}{20}\right) \cos\left(\frac{1}{10}i - \frac{1}{20}\right)\right)^2}$$

$$+ \frac{1}{30} \sum_{i=1}^9 \sqrt{1 + \left(\sin\left(\frac{1}{10}i\right) + \frac{1}{10}i \cos\left(\frac{1}{10}i\right)\right)^2}$$

```
> S:=evalf(e2);
```

$$S := 1.351336227$$

(2.1.3.6)

⌊ ⌊ ⌊ The results L and S are almost indistinguishable.
⌊ >